# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=lulzbuster
pkgver=1.3.3
pkgrel=1
groups=('blackarch' 'blackarch-webapp' 'blackarch-scanner' 'blackarch-recon')
pkgdesc='A very fast and smart web directory and file enumeration tool written in C.'
url='http://www.nullsecurity.net/tools/scanner.html'
license=('custom:unknown')
arch=('x86_64' 'aarch64')
depends=('curl')
source=("https://github.com/nullsecuritynet/tools/raw/main/scanner/lulzbuster/release/$pkgname-$pkgver.tar.gz")
sha512sums=('0e63b008a7b139671cac40fdf91f9985ab88225dbcbfbfbeef51b154ee2116cae72f44efabb2394df69ece151a5a02647c9e107d38fecc82186ab110091cd365')

prepare() {
  cd "$pkgname-$pkgver"

  sed -i 's|/usr/local|/usr|g' inc/opts.h
}

build() {
  cd "$pkgname-$pkgver"

  make clean
  make lulzbuster
}

package() {
  cd "$pkgname-$pkgver"

  make DESTDIR="$pkgdir/usr" install
}

